
3
Applications avancées
Ce chapitre présente des outils un peu plus complexes, tant du point de vue de Scilab
que du traitement du signal. Les exemples se fondent sur les principales méthodes de
décomposition temps-échelle ou temps-fréquence. Les deux premiers mettent en œuvre
la transformée en ondelettes continue et discrète. L'utilisation des ondelettes requiert
quelques fonctions spécifiques incluses dans le module externe SWT. La décomposition
en ondelettes est exposée à partir de la convolution déjà vue au chapitre Applications.

Les différents programmes présentés tirent parti des fonctionnalités graphiques et de
création d'interfaces de Scilab, facilitant les simulations et l'analyse des résultats. Vous
verrez ainsi, avec l'exemple d'un programme permettant d'atténuer le bruit, comment
créer une interface composée de boutons et de listes de choix et récupérer un événement
de type clic pour déclencher une action particulière par l'intermédiaire de fonctions dites
callback.

Les deux derniers exemples utilisent les transformées de Fourier à court terme, de Wi-
gner-Ville et de Choi-Williams. Celles-ci sont comparées en simulant l'effet Doppler d'une
source en mouvement émettant un son et d'un observateur lui aussi en mouvement écou-
tant ce son émis. Les trajectoires sont construites à partir de points de contrôles sélec-
tionnés par l'utilisateur. L'utilisateur peut alors choisir entre une trajectoire composée de
segments de droite ou bien lissée selon la technique des descripteurs de Fourier.

Pour finir, nous procéderons à la séparation aveugle de trois signaux mélangés en utili-
sant l'hypothèse de parcimonie.

3.1. Transformée en ondelettes continue

Soit f une fonction de carré intégrable. La transformée en ondelettes continue de la
fonction f par rapport à l'ondelette  est égal à :

http://d-booker.jo.my/sci-swt

où la fonction  est appelée ondelette mère. La variable a, facteur d'échelle, joue
le rôle d'une fréquence et b d'une translation dans le temps.

La transformée en ondelettes peut être considérée comme une intercorrélation entre

f(t) et  ou bien une convolution entre f(t) et .

Les variables a et b étant des nombres réels, la transformée en ondelettes est continue.

La représentation graphique de la transformée en ondelettes est alors une image
avec en abscisse le temps et en ordonnée le facteur d'échelle.

Nous pouvons estimer la valeur de l'intégrale  en utilisant la fonction
d'intégration de Scilab inttrap.

function F=TOC(f,psi,A,B)
// f expression littérale de la fonction
// psi expression littérale de psi
// A facteur d'échelle 
// B valeur de temps
    // Le résultat est une matrice comportant autant de lignes
    // que de facteurs d'échelle A et autant de colonnes
    // que de valeurs de temps
    F=zeros(length(A),length(B));
    [nbl nbc]=size(F);
    // Valeur du facteur d'échelle
    for i=1:nbl
        a=A(i);
        // Valeur de temps
        for j=1:nbc
            b=B(j);
            // Intégration du produit de l'ondelette par la 
            // fonction au point a,b
            w=inttrap(t,eval(f).*eval(psi));
            F(i,j)=w;
        end
    end
endfunction

Le premier paramètre de cette fonction est l'expression littérale de la fonction dont la
transformée en ondelettes est recherchée. Le deuxième paramètre est l'expression litté-
rale de l'ondelette. Les troisième et quatrième paramètres sont respectivement le facteur
d'échelle et le décalage temporel.

L'ondelette mère  choisie ici est le chapeau mexicain :

La fonction OT_Mexicain suivante donne les versions dilatées d'un facteur d'échelle a et
décalées d'une quantité b :

function s=OT_Mexicain(t,a,b)
    s=2/(%pi^0.25*sqrt(3)).*(1-((t-b)/(a)).^2).*exp(-1/2*((t-b)/
(a)).^2)/sqrt(a);
endfunction

Pour illustrer l'appel à la fonction TOC, une fonction définie dans la variable fct est
étudiée entre 0 et D utilisant les fonctions Porte  et Triangle .

function y=Porte(x) 
    y=zeros(size(t,1),size(t,2));
    ind=find(x>-0.5 & x<0.50);
    if length(ind)>0
        y(ind)=1;
    end
endfunction

function y=Triangle(t) 
    y=zeros(size(t,1),size(t,2));
    ind=find(t>-1/2 & t<=0);
    y(ind)=1+2*t(ind);
    ind=find(t>0 & t<=1/2);
    y(ind)=1-2*t(ind);
endfunction

// Échelle du signal
p=6
// Nombre d'échantillons
N=2^p;
// Durée du signal
D=1;
// Expression littérale de la fonction
fct='Porte((t-D/10)/(D/10))+Triangle((t-D/3)/(D/3))+Porte((t-3*D/4)/
(D/2)).*sin(2*%pi/(D/4)*t)+0.2*Porte((t-7*D/8)/(D/4)).*sin(2*%pi/
(D/11.14)*t)';
t=0:1/203:D;

Nous allons estimer la transformée en ondelettes continue pour les valeurs de temps et
d'échelle contenues dans les vecteurs b et a :

t=0:1/203:D;
// Valeurs de temps où la transformée en ondelettes sera calculée
b=t;
// Valeurs d'échelle où la transformée en ondelettes sera calculée
a=2.^(0:0.05:p)/N;

La transformée en ondelettes est calculée en appelant la fonction TOC avec comme pa-
ramètres la fonction définie dans fct, l'ondelette mère, les valeurs de temps et d'échelle :

// Calcul de la transformée en ondelettes continue
w=TOC(fct,'OT_Mexicain(t,a,b)',a,b);

Le résultat de ces transformées sont des matrices. Nous utilisons la fonction Scilab Mat-
plot pour les représenter sur la fenêtre graphique n°1 (voir Figure 3.1). Les résultats
sont normalisés entre 0 et 255 pour utiliser la palette Scilab graycolormap avec 256
couleurs.

Note > Pour plus d'informations sur l'emploi des palettes de couleurs Scilab, vous pouvez vous reporter
au chapitre Tracés en trois dimensions du module I. Les fondamentaux.

// représentation graphique du résultat sous forme d'image
figure(1);clf();
// Choix d'une palette de 256 couleurs
nbCouleur=256;
// Palette Dégradé de gris
p = graycolormap(nbCouleur);
// normalisation du résultat entre 0 et nbCouleur
// avec maximum en noir et minimum en blanc
ww=((w-max(w))/(min(w)-max(w)))*(nbCouleur-1);
// Tracé de l'image de la matrice
Matplot(fix(ww));
// Changement de la palette
f=gcf();f.color_map=p;f.background=nbCouleur;

Nous graduons les axes en modifiant leurs propriétés récupérées avec la fonction Scilab
gca.

aa=gca();
// Légende du graphique sur X
legGradx=[];
// On veut 6 graduations
nbGradX=6
// Position des graduations
px=[];
for i=0:nbGradX-1
         ind=min(fix(i*size(w,2)/(nbGradX-1))+1,size(w,2));
         px=[px ind];
         legGradx=[legGradx; string(fix(100*b(ind))/100)];
end
// Légende du graphique sur Y
legGrady=[];
// On veut 10 graduations sur Y
nbGradY=10;
py=[];
for i=0:nbGradY-1
         ind=min(fix(i*size(w,1)/(nbGradY-1))+1,size(w,1));

         py=[py ind];
         legGrady=[ string(round(100*log(a(ind))/log(2))/100);legGrady];
end
// Modification des propriétes des axes
aa.x_ticks = tlist(["ticks", "locations", "labels"], px', legGradx);
aa.y_ticks = tlist(["ticks", "locations", "labels"], py', legGrady);
aa.x_location = 'bottom';
aa.y_location = 'left';
xlabel('$ Temps(s) $')
ylabel('$ log_2(a) $')

Figure 3.1 : Transformée en ondelettes continue (Fenêtre graphique n°1)

Sur la fenêtre graphique n°2, nous traçons la fonction sur le graphe du haut et sa trans-
formée en ondelettes au niveau 0 et 6 sur le graphe du bas (voir Figure 3.2).

figure(2);
clf();
// Tracé de la fonction dont on veut l'ondelette
subplot(2,1,1)
plot(t,eval(fct),'r')
xlabel('$ Temps(s) $')
ylabel('$ f(t)$')
xgrid();
// Tracé de transformée en ondelettes au facteur d'échelle 0 et 6
subplot(2,1,2)
plot(b,w(1,:),'b-')
plot(b,w(length(a),:),'r-')
xlabel('$ Temps(s) $')
ylabel('$ W_{f} $')
xgrid();

La fonction cwt du module externe Scilab Wavelet Toolbox (SWT) calcule la transformée
en ondelettes continue pour des facteurs d'échelle entiers. Le premier argument de la
fonction cwt est le signal, le deuxième les facteurs d'échelle et le dernier le nom de
l'ondelette. Les résultats sont tracés sur la même fenêtre graphique n°2 que pour le
calcul précédent (voir Figure 3.2).

Note > La Scilab Wavelet Toolbox propose pour Scilab des fonctionnalités similaires à la Matlab Wavelet
Toolbox pour MATLAB. Pour l'installer, lancez le gestionnaire de module ATOMS depuis le menu APPLI-
CATIONS, et dans la liste des modules proposés, sélectionnez SIGNAL PROCESSING/SCILAB WAVELET TOOLBOX.
N'oubliez pas de redémarrer Scilab !

Figure 3.2 : La fonction et ses coefficients en ondelettes au niveau 0 et 6

// Calcul identique avec les fonctions du module SWT
nn=1:N;
// Période d'échantillonnage
Te=1/64;
te=(0:N-1)*Te;   
t=te;
// Évaluation de la fonction
ye=eval(fct);
// Calcul de l'ondelette continue
c=cwt(ye,nn,'mexh');
// Représentation graphique sur la même fenêtre graphique n°2
subplot(2,1,1) 
plot(te,ye,'+')
xgrid();
subplot(2,1,2)

http://d-booker.jo.my/sci-swt

plot(te,c(1,:)*sqrt(Te),'b+')
plot(te,c( length(nn),:)*sqrt(Te),'ro')
xgrid();
legend('$ W_f{1,t} $','$ W_f{Te,t} $','$ W_f{1,t}(cwt) $','$ W_f{Te,t}
(cwt) $');

La fonction cwtplot du module SWT permet de représenter la transformée en onde-
lettes sous forme d'images avec comme premier paramètre les coefficients d'ondelettes,
comme deuxième paramètre les échelles des coefficients, comme troisième paramètre
une valeur logique indiquant si la palette est visible et en dernier paramètre l'indice
de la fenêtre graphique Scilab. Pour obtenir la même image que précédemment avec
Matplot, nous devons modifier les facteurs d'échelle et la méthode de normalisation du
résultat de la transformée en ondelettes continue. La discrétisation du signal étant faite
avec une période de Te, les facteurs d'échelle sont égaux à (1:N)*Te.

Pour obtenir une équivalence entre les représentations graphiques obtenues par Mat-
plot et cwtplot, la ligne ww=((w-max(w))/(min(w)-max(w)))*(nbCouleur-1) doit être
remplacée par ww=abs(w)/max(abs(w))*nbCouleur.

// Représentation graphique du résultat sous forme d'image
// en utilisant cwtplot
figure(3);clf();
cwtplot(c,nn,%f,3);
f=gcf();
p = graycolormap(256);
f.color_map=p;f.background=256;

Note > Vous pouvez retrouver le code complet de ce programme dans le fichier TOC.sce.

3.2. Transformée en ondelettes discrète

La transformée en ondelettes continue  est une fonction continue du fac-
teur d'échelle a et de translation b. La transformée en ondelettes discrète est une
analyse multirésolution fonction de ces deux mêmes paramètres avec  et

 où n et m sont des entiers. Lorsque  et , la transformée est
dite dyadique. La transformée en ondelettes discrètes étant calculée pour un nombre
fini de valeurs du facteur d'échelle, il faut introduire une information complémentaire
pour assurer la reconstruction de la fonction f à partir de sa transformée. C'est le
rôle de la fonction d'échelle.

L'algorithme de décomposition en ondelettes peut se résumer en trois étapes :


